Résolution RP-4

© Pierre Lantagne
Enseignant retraité du Collège de Maisonneuve

* Ce document Maple est exécutable avec la version 2020.2

Initialisation

```
> restart;
> with(Physics[Vectors]):
  with(LinearAlgebra):
  with(plots,display,pointplot3d,setoptions3d):
  setoptions3d(labels=[x,y,z],axis[3]=[color="Niagara 1"],
               tickmarks=[4,4,4],lightmodel=none,
               axesfont=[TIMES,ROMAN,10],
               labelfont=[TIMES,ROMAN,16]);
```

No. 1

Soit les plans d'équation $\Pi_1\ : 2x + y + z - 1 = 0$,

$\Pi_2\ : 6x + y + z - 3 = 0$ et

$\Pi_3\ : x + y - z = 0$.

a) Sachant que ces trois plans se coupent en un point, obtenons immédiatement les coordonnées du point d'intersection P.

```
> Éq1:=2*x+y+z-1 = 0:
  Éq2:=6*x+y+z-3 = 0:
  Éq3:=x+y-z = 0:
  Système:={Éq1,Éq2,Éq3};
```

$$Système := \{x + y - z = 0, 2x + y + z - 1 = 0, 6x + y + z - 3 = 0\} \quad \textbf{(2.1)}$$

```
> solve(Système,{x,y,z}) assuming z::real;
```

$$\left\{x = \frac{1}{2}, y = -\frac{1}{4}, z = \frac{1}{4}\right\} \quad \textbf{(2.2)}$$

Les coordonnées du point d'intersection sont $P\left(\frac{1}{2}, -\frac{1}{4}, \frac{1}{4}\right)$.

b) Illustrons dans un même graphique ces trois plans ainsi que le point d'intersection P.

```
> Point:=pointplot3d([1/2,-1/4,1/4],symbol=solidsphere,symbolsize=
  15,color=black):
> Plan1:=plot3d([x,y,1-2*x-y],x=-3..3,y=-3..3,color="Bright 1",
           style=patch,grid=[2,2]):
  Plan2:=plot3d([x,y,3-6*x-y],x=-3..3,y=-3..3,color="Bright 5",
            style=patch,grid=[2,2]):
  Plan3:=plot3d([x,y,x+y],x=-3..3,y=-3..3,color="Bright 6",
            style=patch,grid=[2,2]):
```

```
> display({Point,Plan||(1..3)},axes=framed,orientation=[50,75],view=
  [-3..3,-3..3,-3..4]);
```



c) Résolvons les trois sous-systèmes de deux équations pris à la fois.

```
> D1:=solve({Éq1,Éq2}) assuming z::real;
  D2:=solve({Éq1,Éq3}) assuming z::real;
  D3:=solve({Éq2,Éq3}) assuming z::real;
```

$$D1 := \left\{x = \frac{1}{2}, y = -z, z = z\right\}$$

$$D2 := \{x = 1 - 2z, y = -1 + 3z, z = z\}$$

$$D3 := \left\{x = \frac{3}{5} - \frac{2z}{5}, y = -\frac{3}{5} + \frac{7z}{5}, z = z\right\} \quad \textbf{(2.3)}$$

En confirmité avec les choix de Maple pour la désignation des variables libres, les équations paramétriques des trois droite sont:

Δ_1: $x = \frac{1}{2}$, $y = -t$, $z = t$, $t \in \mathbb{R}$. , Δ_2: $x = 1 - 2t$, $y = -1 + 3t$, $z = t$, $t \in \mathbb{R}$. et Δ_3: $x = \frac{3}{5} - \frac{2t}{5}$, $y = -\frac{3}{5} + \frac{7t}{5}$, $z = t$, $t \in \mathbb{R}$.

d) Dans un même graphique, superposons ces trois droites.

```
> Delta1:=plot3d([1/2,t,-t],t=-3..3,s=-5..5,thickness=2,color="MapleV 11",style=line):
  Delta2:=plot3d([1-2*t,-1+3*t,t],t=-3..3,s=-5..5,thickness=2,color="MapleV 6",style=line):
  Delta3:=plot3d([3/5-2/5*t,-3/5+7/5*t,t],t=-3..3,s=-5..5,thickness=2,color="MapleV 24",style=line):
```

```
> display({Delta||(1..3),Point},axes=framed,orientation=[25,70],view=[-3..3,-3..3,-3..3]);
```



e)

```
> display({Point,Plan||(1..3),Delta||(1..3)},axes=framed,labels=[x,y,z],lightmodel=none,
         view=[-3..3,-3..3,-10..10],orientation=[75,80]);
```



No. 2

Soit le plan Π: $2x - 3y + 6z - 4 = 0$ et la droite Δ: $\dfrac{x-2}{4} = \dfrac{y+6}{-1} = \dfrac{z-4}{3}$.

a) La droite Δ coupant le plan Π, trouvons les coordonnées du point de percé.

Déduisons d'abord les équations paramétriques de la droite Δ: $x = 2 + 4t$

$$y = -6 - t$$
$$z = 4 + 3t, \quad t \in \mathbb{R}.$$

Ainsi, substituons x, y et z dans l'équation du plan et résolvons pour t, l'équation obtenue.

```
> PI:=2*x-3*y+6*z-4 = 0;
  Point:=[2+4*t,-6-t,4+3*t];
```

$$\Pi := 2x - 3y + 6z - 4 = 0$$

$$Point := [2 + 4t, -6 - t, 4 + 3t] \quad \textbf{(3.1)}$$

```
> Valeur_de_t:=solve(subs([x = 2+4*t,y = -6-t,z = 4+3*t],PI) ,{t});
```

$$Valeur\_de\_t := \left\{ t = -\frac{42}{29} \right\} \quad \textbf{(3.2)}$$

```
> Point_de_percé:=eval(Point,Valeur_de_t);
```

(3.3)

$$Point\_de\_percé := \left[-\frac{110}{29}, -\frac{132}{29}, -\frac{10}{29} \right] \tag{3.3}$$

Les coordonnées du point de percé sont $\left[-\frac{110}{29}, -\frac{132}{29}, -\frac{10}{29} \right]$.

b) Illustrons, dans un même graphique, le plan Π, la droite Δ et le point de percée.

```
> Droite:=plot3d([2+4*t,-6-t,4+3*t],s=-3..3,t=-2..-1,thickness=2,color=
  "Niagara 1",style=line):
```

```
> Plan:=plot3d([s,t,(-2*s+3*t+4)/6],s=-7..2,t=-7..2,style=Surface,grid=
  [2,2],color="MapleV 16"):
```

```
> Point:=pointplot3d([-110/29, -132/29, -10/29],symbol=solidsphere,
  symbolsize=15,color=navy):
```

```
> display(Droite,Plan,Point,axes=framed,orientation=[130,45]);
```



c) Déterminons l'angle de percée. Obtenons en premier l'un des angles supplémentaires formés par un vecteur normal au plan Π et un vecteur directeur de la droite Δ.

```
> n_:=<2|-3|6>;
  d_:=<4|-1|3>;
```

$$\vec{n} := \begin{bmatrix} 2 & -3 & 6 \end{bmatrix}$$

(3.4)

$$\vec{d} := \begin{bmatrix} 4 & -1 & 3 \end{bmatrix} \quad \mathbf{(3.4)}$$

```
> Angle:=evalf(convert(arccos(n_ . d_ /VectorNorm(n_,Euclidean)
  /VectorNorm(d_,Euclidean)),degrees));
```

$$Angle := 35.6609421700\,degrees \quad \mathbf{(3.5)}$$

L'angle de percée est donc l'angle complémentaire à l'angle de 35.6609421700 *degrees*.

```
> Angle_percée:=90*degrees-Angle;
```

$$Angle\_percée := 54.3390578300\,degrees \quad \mathbf{(3.6)}$$

No. 3

Soit le point $P_0(6, 3, -2)$ et la droite Δ: $\dfrac{x-2}{5} = \dfrac{y+1}{2} = z - 3$.

a) Trouvons la distance du point P_0 à la droite Δ.

D'abord, déterminons un vecteur P_0P où $P(2, -1, 3)$ est un point de la droite Δ. Ensuite, évaluons la norme du produit vectoriel de P_0P avec un vecteur directeur de la droite Δ divisée par la norme du vecteur directeur.

```
> PoP_:=<2-6|-1-3|3-(-2)>;
  vd_:=<5|2|1>;
```

$$\overrightarrow{PoP} := \begin{bmatrix} -4 & -4 & 5 \end{bmatrix}$$

$$\overrightarrow{vd} := \begin{bmatrix} 5 & 2 & 1 \end{bmatrix} \quad \mathbf{(4.1)}$$

```
> VectorNorm(PoP_ &x vd_,Euclidean)/VectorNorm(vd_,Euclidean);
```

$$\frac{\sqrt{1181}\,\sqrt{30}}{30} \quad \mathbf{(4.2)}$$

La distance cherchée est donc de $\dfrac{\sqrt{1181}\,\sqrt{30}}{30}$ unités.

b) Illustrons cette distance avec un segment de droite reliant perpendiculairement le point P_0 à la droite Δ.

Déterminons d'abord les coordonnées du pied de la perpendiculaire abaissé du point P_0 à la droite Δ.
Soit un point $P(2+5t, -1+2t, 3+t)$ un point quelconque de la droite Δ et $vd\_ = \langle 5,2,1 \rangle$ une vecteur directeur de la droite Δ. Considérons le vecteur
$P_0P = (6-(2+5t), 3-(-1+2t), -2-(3+t))$ et résolvons, pour t, le produit scalaire $P_0P \cdot vd\_ = 0$.

```
> PointR:=[6,3,-2];
  PointP:=[2+5*t, -1+2*t, 3+t];
```

$$PointR := [6, 3, -2]$$

$$PointP := [2+5t, -1+2t, 3+t] \quad \mathbf{(4.3)}$$

```
> v_:=convert(PointP-PointR,Vector[row]);
  vd_:=<5|2|1>;
  Éq:=v_ . vd_;
```

$$\vec{v} := \begin{bmatrix} -4+5t & -4+2t & 5+t \end{bmatrix}$$

$$\overrightarrow{vd} := \begin{bmatrix} 5 & 2 & 1 \end{bmatrix}$$

$$\acute{E}q := -23 + 30\,t \tag{4.4}$$

```
> solve(Éq,{t});
```

$$\left\{ t = \frac{23}{30} \right\} \tag{4.5}$$

Les coordonnées du pied de la perpendiculaire sont

```
> Pied:=eval([2+5*t,-1+2*t,3+t],t=23/30);
```

$$Pied := \left[\frac{35}{6}, \frac{8}{15}, \frac{113}{30} \right] \tag{4.6}$$

```
> Point_P:=pointplot3d([6,3,-2],symbol=solidsphere,color="Resene 189",
  symbolsize=15):
  Point_R:=pointplot3d([35/6, 8/15, 113/30],symbol=solidsphere,color=
  "Resene 169",symbolsize=15):
> Droite:=plot3d([2+5*t,-1+2*t,3+t],s=-3..3,t=-3..3,thickness=2,color=
  "Resene 42",style=line):
> Lieu:=expand(PointR+t*(Pied-PointR));
```

$$Lieu := \left[-\frac{t}{6} + 6, -\frac{37\,t}{15} + 3, \frac{173\,t}{30} - 2 \right] \tag{4.7}$$

```
> Segment:=plot3d(Lieu,t=0..1,s=-5..5,thickness=2,color="Resene 173",
  style=line):
  display({Point_P,Point_R,Droite,Segment},axes=framed,orientation=[60,
  65],
          view=[0..10,-5..5,-3..6]);
```


5
2.5
z
0
-2.5
-5
0
y
10
5
x
0